Newsgroup: it.comp.appl.access
Titolo: leggere le "proprietà database"

ciao
da access apro menu > file > proprietą Database ed in Personalizza posso
aggiungere delle "proprietą" con relativo valore, da codice come posso
leggere tale valore ?
ciao e grazie


Vatti a leggere "Insieme Properties" sulla guida in linea.
SunRex

 leggere tale valore ?
Ciao, Liones.
        Access.Application.CurrentDb _
        .Containers!Databases.Documents!UserDefined _
        .Properties![Nome della proprietą].Value
    Sto testando il metodo seguente. Se vuoi provarlo anche tu fammi sapere
se hai problemi, grazie. In pratica uso le routine Property non in un Modulo
di classe ma in un Modulo standard. In modo da poterle usare nel codice per
esempio cosi':
    Assegnazione di valore alla proprieta' del database:
        DurataInMillisecondi = 3000
    Assegnazione del valore della proprieta' del database ad una variabile:
        Dim lngDurata As Long
        lngDurata = DurataInMillisecondi
    Se la proprieta' del database non esiste ancora, viene creata con le
impostazioni predefinite nel progetto visual basic.
    Per ogni proprieta' che voglio gestire nel progetto visual basic copio e
in collo le tre routine (Get, Let e _Error), trovo/sostituisco il nome, i
valori predefiniti ed il tipo. Se qualcosa non e' chiaro... mi trovi qui.
' Modulo: MUsrProperties
' ============================================================
' Maurizio Borrelli
' ------------------------------------------------------------
  Option Base 0
  Option Compare Database
  Option Explicit
  Const mc_strMyName As String = "MUsrProperties"
' ============================================================
' ============================================================
  Public Property Get DurataInMillisecondi( _
  ) As Long
' ------------------------------------------------------------
  Const c_strMyName _
              As String = "DurataInMillisecondi [PropertyGet]"
  On Error GoTo ErrorHandler
  DurataInMillisecondi = Access.Application.CurrentDb _
    .Containers!Databases.Documents!UserDefined _
    .Properties!DurataInMillisecondi.Value
  Exit Property
ErrorHandler:
  DurataInMillisecondi = DurataInMillisecondi_Error(Err)
  End Property
' ============================================================
  Public Property Let DurataInMillisecondi( _
    ByVal NewValue As Long)
' ------------------------------------------------------------
  Const c_strMyName _
              As String = "DurataInMillisecondi [PropertyLet]"
  On Error GoTo ErrorHandler
  Access.Application.CurrentDb _
    .Containers!Databases.Documents!UserDefined _
    .Properties!DurataInMillisecondi.Value = NewValue
  Exit Property
ErrorHandler:
  Call DurataInMillisecondi_Error(Err, NewValue)
  End Property
' ============================================================
  Private Function DurataInMillisecondi_Error( _
    ByVal ObjectError As VBA.ErrObject _
  , Optional ByVal NewValue As Long = 2000 _
  ) As Long
' ------------------------------------------------------------
  Const c_strMyName   As String = "DurataInMillisecondi"
  Const c_lngType     As Long = DAO.dbLong
  Dim objDaoDbs       As DAO.Database
  Dim objDaoProperty  As DAO.Property
  If (ObjectError.Number = 3270) Then
    Set objDaoDbs = Access.Application.CurrentDb
    Set objDaoProperty = objDaoDbs _
      .Containers!Databases.Documents!UserDefined _
      .CreateProperty(c_strMyName, c_lngType, NewValue)
    objDaoDbs.Containers!Databases.Documents!UserDefined _
      .Properties.Append objDaoProperty
    DurataInMillisecondi_Error = objDaoDbs _
      .Containers!Databases.Documents!UserDefined _
      .Properties.Item(c_strMyName).Value
    Set objDaoProperty = Nothing
    Set objDaoDbs = Nothing
  Else
    Debug.Print ErrMsg(Err, mc_strMyName, c_strMyName)
  End If
  End Function
' ============================================================
' Modulo: MGlobal
' ============================================================
' Maurizio Borrelli
' ------------------------------------------------------------
  Option Base 0
  Option Compare Database
  Option Explicit
  Const mc_strMyName As String = "MGlobal"
' ============================================================
' ============================================================
  Public Function ErrMsg( _
    ByVal ErrorObject As VBA.ErrObject _
  , ByVal ModuleName As String _
  , ByVal ProcedureName As String _
  ) As String
' ------------------------------------------------------------
  Const c_strMyName As String = "ErrMsg"
  Dim strResult As String
  Dim strTime   As String
  Dim strIndent As String
  strTime = Format$(Now, "hh:nn:ss")
  strIndent = vbNewLine & String$(Len(strTime), " ")
  With ErrorObject
    strResult = strTime _
      & ":" & ModuleName & "." & ProcedureName _
      & strIndent & ":" & .Source _
      & strIndent & ":ERR#" & CStr(.Number) _
      & ":" & .Description
  End With
  ErrMsg = strResult
  End Function
' ============================================================
-- 
    (Facci sapere se ed eventualmente come hai risolto. Grazie.)
Ciao :o)
Maurizio Borrelli, Microsoft MVP Access
--------
?SPQR(C)
 X
--------

     Sto testando il metodo seguente. Se vuoi provarlo anche tu fammi
sapere
quando ho 2 minuti (praticamente mai)
di nuovo grazie

 quando ho 2 minuti (praticamente mai)
Ciao, Liones.
    Beh... Io li ho trovati per risponderti!...
-- 



sigh
ti ringrazio moltissimo
vediamo
oggi è il 20/01/04, domani terminerò un lavoro che doveva finire il 31/12/03
finisco di lavora alle 18,30
un'ora di macchina
lettura di: it.comp.appl.access, it.comp.lang.delphi (da alcuni mesi cerco
di impare delphi (bellissimo)), it.comp.lang.visual-basic,
microsoft.public.it.sql (adp), etc
"devo" imparare net
etc.
(il 90% nel mio termpo libero)
mi sono chiesto più volte come fate a trovare il tempo per rispondere ai ng
di nuovo grazie
ciao

 mi sono chiesto piů volte come fate a trovare il tempo per rispondere ai
ng
Ciao, Liones.
    Da queste parti siamo in molti che abbiamo smesso di farci certe domande
perche' abbiamo paura della risposta.
    :-))))
